1. (15) A palindrome is a word or phrase that is spelled the same way forward and backward. For example, the name of the Swedish singing group ABBA is a palindrome.

Assume that you have a form with one text box and a command button. Write a click event for the command button that determines if the word in the text box is or is not a palindrome. The first time the user clicks the command button, the first and last characters of the word should be checked. The second click should check the second and second to last characters. This pattern should continue until the word is determined to be (or not to be) a palindrome.

The click event should report via a message box as soon as it can tell if the word is not a palindrome. That is, if this conclusion can be drawn after the first click, the routine should display the message box.

After the status of the word is determined (palindrome or not palindrome), the user should be able to enter a new word into the text box and then determine its palindrome status.

2. (10) Evaluate each of the following VB expressions. Assume that all the variables are declared as type Single, and that \( A = 10 \), \( B = 2 \), \( C = 5 \), \( D = 25 \), and \( E = 3 \).

   a. \( A + B \times C + D \)

   b. \( A^2 + (D / C)^E \)

   c. \( (A \div E) \times (D \mod A) \)

   d. \( A / B \times C / D \)

   e. \( B^3 \times B / B \)
3. (12) You are given the following click event and three procedures.

```vba
Option Explicit 'first two lines are in the form's General Declarations section
Dim X As Integer

Private Sub cmdStart_Click()
    Dim A As Integer
    Dim B As Integer
    Dim C As Integer
    X = 100
    A = 10
    B = 25
    C = 50
    subA A, X
    Form1.Print A, B, C, X
    subB B, C
    Form1.Print A, B, C, X
    subC A, B, C
    Form1.Print A, B, C, X
End Sub

Private Sub subA(X As Integer, ByVal Y As Single)
    Dim A As Integer
    A = X
    X = 13
    Y = 101
End Sub

Private Sub subB(A As Integer, B As Integer)
    Dim X As Integer
    X = A
    B = 18
    A = 33
End Sub

Private Sub subC(S As Integer, T As Integer, U As Integer)
    X = 44
    S = T
    subA T, U
    subB S, T
End Sub
```

Determine what will be displayed by the three Form1.Print statements when the user clicks on cmdStart_Click(). Be sure to clearly identify and show the values and indicate which Form1.Print statement displayed them.
4. (8) A company uses the following table to determine the shipping charges for its products:

<table>
<thead>
<tr>
<th>Weight</th>
<th>&lt; 500 miles</th>
<th>≥ 500 miles</th>
</tr>
</thead>
<tbody>
<tr>
<td>&lt; 5 pounds</td>
<td>$5.00</td>
<td>$7.25</td>
</tr>
<tr>
<td>≥ 5 pounds</td>
<td>$6.50</td>
<td>$10.50</td>
</tr>
</tbody>
</table>

Write a VB function that determines the appropriate shipping charge. Assume the function is passed the distance and weight as integers and that it returns the shipping charge as a currency type.

5. (8) You have a text box named txtPartNumber. You want to make sure that when the user finishes typing in the box (as indicated by the user clicking on any other control), the text box has at least three characters in it. If it does not, display a message to the user and set the focus back to the text box.

Write the code for the appropriate event procedure to accomplish this.
Sample Answers:

1.

```vbnet
Private Sub cmdCheckLetter_Click()
    Static letterPos As Integer
    Dim word As String
    letterPos = letterPos + 1
    word = txtWord.Text
    If letterPos >= (Len(word) - letterPos + 1) Then
        MsgBox "It's a palindrome"
        letterPos = 0
        Exit Sub
    End If
    If Mid$(word, letterPos, 1) <> Mid$(word, Len(word) - letterPos + 1, 1) Then
        MsgBox "Not a palindrome"
        letterPos = 0
    End If
End Sub
```

2. a. 45  
   b. 225  
   c. 15  
   d. 1  
   e. 4

3. First Form 1.Print  A=13  B=25  C=50  X=100
   Second Form 1.Print  A=13  B=33  C=18  X=100
   Third Form 1.Print  A=33  B=18  C=18  X=44
4. 

    Private Function shippingCharge(ByVal dist As Integer, ByVal weight As Integer) _
          As Currency
          Select Case dist
              Case Is < 500
                  Select Case weight
                      Case Is < 5
                          shippingCharge = 5
                      Case Is >= 5
                          shippingCharge = 6.5
                  End Select
              Case Is >= 500
                  Select Case weight
                      Case Is < 5
                          shippingCharge = 7.25
                      Case Is >= 5
                          shippingCharge = 10.5
                  End Select
          End Select
    End Function

5. 

    Private Sub txtPartNumber_LostFocus()
        If Len(txtPartNumber.Text) < 3 Then
            MsgBox "You must supply at least 3 characters"
            txtPartNumber.SetFocus
        End If
    End Sub